

Як програмно відкрити форму звіту з встановленим відбором по полю/параметру СКД?

Для реалізації задачі будемо використовувати параметри форми. Параметри форми необхідні для організації зв'язку між різними формами та управління функціональністю форми при її відкритті.

Параметри може створювати розробник на відповідній закладці «Параметри» керованої форми. Але також формою автоматично надаються параметри, які обумовлені типом основного реквізиту. Нас будуть цікавити саме параметри у керованій форми **звіту**. Їх опис можна знайти у *Синтаксис-помічнику* (рис. 1).




Рис. 1. Параметры керованої форми звіту у *Синтаксис-помічнику*

Серед цих параметрів, що надаються платформою автоматично, детально розглянемо опис параметру «Отбор» та «СформироватьПриОткрытии».

- **Отбор (Filter)**

- ✓ Тип: *Структура*.

- ✓ Опис: Використовується для установки відбору і параметрів даних у формі звіту. Ключ елемента структуры містить ім'я **поля** або **параметра**, на яке потрібно накладати відбір. Значення елемента структуры містить значення відбору.

*Якщо є **параметр** системи компоновки даних з ім'ям, що відповідає імені ключа структуры, то значення буде встановлено йому. Якщо параметра немає, але є **поле**, то буде доданий відбір на це поле.*

Якщо значення є масивом або списком значень з декількома елементами, то відбір буде застосовуватися з видом порівняння: **ВСписке**.

- **СформироватьПриОткрытии (GenerateOnOpen)**

- ✓ Тип: **Булево**.

- ✓ Опис: Вказує необхідність формування звіту при відкритті. **Истина** - звіт буде сформовано відразу на сервері, після обробника події **ПриСозданииНаСервере**.

Продemonструємо приклад застосування наведених параметрів. Для цього реалізуємо наступну задачу.

Завдання. Реалізувати можливість програмно відкрити і сформувати звіт «ЦеныПоставщиков» з відбором по одному або декільком контрагентам.

Реалізація завдання.

&НаКлиенте

Процедура ОтчетПоВыбраннымКонтрагентам (МассивКонтрагентов)

//задаємо структуру відбору по полю СКД «Контрагент»

// МассивКонтрагентов – параметр, що містить масив/список посилань на довідник

//Контрагенти

ВыборКонтрагентов = Новый Структура;

ВыборКонтрагентов.Вставить("Контрагент", МассивКонтрагентов);

// задаємо структуру, що містить параметри форми,

// ми використовуємо параметри Отбор та СформироватьПриОткрытии

ПараметрыФормы = Новый Структура("Отбор, СформироватьПриОткрытии");

//задаємо значення ключів структури

ПараметрыФормы.Отбор = ВыборКонтрагентов;

ПараметрыФормы.СформироватьПриОткрытии = Истина;

//відкриваємо форму звіту ЦеныПоставщиков з урахуванням заданих параметрів

ОткрытьФорму("Отчет.ЦеныПоставщиков.Форма", ПараметрыФормы);

КонецПроцедуры

У нашому випадку в основній СКД звіту «ЦеныПоставщиков» присутнє поле, що відповідає імені ключа структури відбору. І оскільки значення ключа у наведеному прикладі є масивом значень, то відбір застосовується з видом порівняння: **ВСписке**, що підтверджується сформованим звітом (рис. 2).

Звернемо увагу, що у заголовку сформованого звіту (рис 2.) буде зазначено «Встановлено додатковий відбір», що означає, що спрацювали фіксовані настройки СКД.

←

→

☆

Ціни постачальників

(Встановлений додатковий відбір)

Сформувати

Вибрати варіант...

Налаштувати...

Дата отчета: ☐ . . : :

Цены поставки

Відбір: Поставщик В списку "ЗАТ "Сонечко"; ТОВ "Зелене світло"

Період	Поставщик	Номенклатура	Артикул	Единица измерения	Цена поставки
3 листопада 2020 р.	ЗАТ "Сонечко"	ELECTROLUX ER 9007 B	000000018	Шт	25 000,00
3 листопада 2020 р.	ЗАТ "Сонечко"	Атлант MXM 1704-00	000000019	Шт	15 000,00
3 листопада 2020 р.	ЗАТ "Сонечко"	BOSCH KGS 3760 IE	000000017	Шт	18 000,00
10 листопада 2020 р.	ТОВ "Зелене світло"	Лак	000000021	Літр	10,00
10 липня 2021 р.	ТОВ "Зелене світло"	Дошка	000000023	Шт	50,00

Рис. 2. Сформований звіт з програмно визначеними параметрами